
.Z80

LD SP,ENDMC+100

LD A,BITB

OUT (PIOAC),A

LD A,01110100B

OUT (PIOAC),A

LD A,(PIO)

OR 01111111B ; ZUSTAND ZX/CPM RETTEN

OUT (PIOAD),A

CA1: CALL CIST

OR A

JR Z,CARA ; TASTEN ABHOLEN

CALL CI

JR CA1

CARA: CALL UPCR

CALL UPCR

CALL PRINT

DEFM " Codecard Leser "

DEFB 0

CALL UPCR

LD HL,CABYTE

LD DE,CABYTE

LD BC,3000H

LD (HL),0FFH

INC DE

LDIR ; PUFFER LOESCHEN

; INTVEKTOR VERBIEGEN

LD HL,(0FF00H) ; INT SIO A

LD (MEMINT),HL

LD HL,INTN

LD (0FF00H),HL

; INT PROGRAMMIERUNG PIO

DI

LD HL,CREAD

LD (0FF10H),HL ; INTVEKTOR

LD A,10H ;INTVEKTOR

OUT (PIOAC),A

LD A,10010111B ; INT STEUER:  ODER/LOW/EI

OUT (PIOAC),A

LD A,11101111B ; INT MASKE

OUT (PIOAC),A

LD A,10000011B ; INT ERLAUBT

OUT (PIOAC),A

EI

LD HL,CABIT

CRHAL: HALT ; SCHLEIFE WARTEN AUF INT

CRHAL0: LD A,L

LD B,0

CRHAL1: DJNZ CRHAL1

CRHAL2: DJNZ CRHAL2

CRHAL3: DJNZ CRHAL3

CP L ; TEST AENDERUNG HL

JR Z,CARE0 ; OHNE AENDERUNG AUSPRUNG

JR CRHAL0

; INT ROUTINE PIO

CREAD: IN A,(PIOAD)

LD (HL),A

INC HL

EI

RETI

; INT SIO ABFANGEN

INTN: DEFB 0CDH ; BEFEHLSCODE CALL

MEMINT: DEFW 0

CALL CI

CP 3

JP Z,CAREN ; AUSSPRUNG BEI ^C

CARE0: LD A,00000011B ; INT PIO VERBOTEN

OUT (PIOAC),A

LD HL,(MEMINT)

LD (0FF00H),HL ; INTVEKTOR CI ORIGINAL

LD HL,CABIT

CARE1: LD A,(HL)

CP 0FFH

JP Z,CARW ; AUSWERTUNG

BIT 5,A

LD (HL),0

JR NZ,CARE2

LD (HL),1

CARE2: LD A,(HL)

ADD A,30H

CALL COA

INC HL

JR CARE1

CARW: LD HL,CABIT

LD DE,CABYTE

CALL UPCR

CARW1: LD A,(HL)

INC HL

CP 0FFH

JP Z,CARA ; FERTIG

CP 1

JR NZ,CARW1

DEC HL

CARW3: LD BC,0500H ; 5 BITS BYTE 0

CARW2: LD A,(HL)

INC HL

CP 0FFH

JP Z,CARK ; FERTIG

ADD A,0FFH ; BEI 1 CARRY , BEI 0 NC

RR C

DJNZ CARW2

LD A,C

RRA

RRA

RRA

AND 1FH ; PARITAETSKONTROLLE

LD C,A

LD A,"?"

JP PE,COUTH2

LD A,C

AND 0FH

ADD A,30H

CP 3AH

JR C,COUTH2

ADD A,7

COUTH2: LD (DE),A

INC DE

CALL COA

JR CARW3 ; NAECHSTES BYTE

CARK: LD A,(CABYTE+33)

CP "D"

CALL Z,CARKS

LD A,(CABYTE+33)

CP "0"

CALL Z,CARKE

JP CARA

CARKS: LD HL,CABYTE+3

CALL UPCR

CALL UPCR

CALL PRINT

DEFM "Sparkasse        Konto-Nummer : "

DEFB 0

LD B,4

CARK1: LD A,(HL)

INC HL

CALL COA

DJNZ CARK1

LD A,"-"

CALL COA 

LD B,8

CARK2: LD A,(HL)

INC HL

CALL COA

DJNZ CARK2

CALL PRINT

DEFB "    Geburtsdatum : ",0

LD B,2

CARK3: LD A,(HL)

INC HL

CALL COA

DJNZ CARK3

LD A,"."

CALL COA

LD B,2

CARK4: LD A,(HL)

INC HL

CALL COA

DJNZ CARK4

LD A,"."

CALL COA

LD B,2

CARK5: LD A,(HL)

INC HL

CALL COA

DJNZ CARK5

CALL UPCR

CALL PRINT

DEFB "                 PKZ          : ",0

LD HL,CABYTE+15

LD B,12

CARK6: LD A,(HL)

INC HL

CALL COA

DJNZ CARK6

CALL UPCR

RET

; EUROCARD

CARKE: LD HL,CABYTE+10

CALL UPCR

CALL UPCR

CALL PRINT

DEFM "Euro Card        Konto-Nummer : "

DEFB 0

LD B,9

CARKE1: LD A,(HL)

INC HL

CALL COA

DJNZ CARKE1

CALL UPCR

RET

JP CARA ; ENDE AUSWERTUNG

CAREN: LD A,00000011B ; INT PIO VERBOTEN

OUT (PIOAC),A

LD HL,(MEMINT)

LD (0FF00H),HL ; INTVEKTOR CI ORIGINAL

JP 0

;-------------------------------

ZEIT: PUSH HL

ZZ1: DEC HL

LD A,H

OR L

JR NZ,ZZ1

POP HL

RET

;------------------------------

; UP SAMMLUNG

;-------------------------------

; INHEX

INHEX: LD A,20H

CALL COA

CALL INL0

ASHEX: CALL FNEXT

CALL CNVBN

LD A,(DE)

RET C ;kein hex

PUSH BC

LD HL,0

ASH1: LD A,(DE)

CALL CNVBN

JR C,ASH4

ASH3: ADD HL,HL

ADD HL,HL

ADD HL,HL

ADD HL,HL

LD B,0

LD C,A

ADD HL,BC

INC DE

JR ASH1

ASH4: POP BC

XOR A

RET

CNVBN: SUB 30H

RET C

CP 10

CCF

RET NC

CP 11H

RET C

CP 17H

CCF

RET C

SUB 7

RET

;

FNEXT: LD A,(DE)

CP 20H

JR NZ,FNEX3

FNEX1: INC DE

LD A,(DE)

CP 20H

JR Z,FNEX1

CP 0DH

JR Z,FNEX4

FNEX2: XOR A

LD A,(DE)

RET

FNEX3: CP 0DH

FNEX4: SCF

RET Z

CCF

CP 2CH

INC DE

JR Z,FNEX2

OR A

DEC DE

RET

;---Eingabe Befehlszeile-------

; 

TXTBU: DS 90

MBULEN: DEFB 4

INLN: CALL PRINT

DEFB 0DH,23H,20H,0

LD B,78

LD DE,TXTBU

LD A,20H

INLN01: LD (DE),A

INC DE

DJNZ INLN01

INL0: LD DE,TXTBU

LD B,78 ;PUFFERLAENGE

INL1: LD A,B

LD (MBULEN),A

INL19: CALL CI

CP 01 ; FN TASTE

JR Z,INL5

CP 08 ; Rueckschritt

JR Z,INL3

CP 0DH

JR Z,INL2

CP 12 ; KURSOR RECHTS

JR Z,INL6

CP 3 ; ^C

JP Z,0

CP 20H

JR C,INL19

LD (DE),A

CALL COA

INC DE

DJNZ INL19

INL3: LD A,(MBULEN)

CP B

JR Z,INL19

DEC DE

INC B

LD A,8; Delete

CALL COA

JR INL19

INL2: LD (DE),A

LD A,0DH

CALL COA

LD DE,TXTBU

RET

INL5: CALL CI ;NUMMER

CALL CI ;ENTER

JR INL19

INL6: LD A,(DE)

CALL COA

INC DE

DJNZ INL19

;-------------------------------

COA: PUSH BC

LD C,A

CALL CO

POP BC

RET

CO: PUSH AF

PUSH HL

PUSH BC

PUSH DE

CALL BIOS+0CH

POP DE

POP BC

POP HL

POP AF

RET

UPCR: PUSH BC

LD C,0AH

CALL CO

LD C,0DH

CALL CO

POP BC

RET

SPACE: PUSH BC

LD C,32

CALL CO

CALL CO

POP BC

RET

CIST: JP BIOS+6

CI: JP BIOS+9

;    PRINT Zeichenkette bis 00

PRINT: EX (SP),HL

PR1: LD A,(HL)

INC HL

OR A

JR Z,PR2

CALL COA

JR PR1

PR2: EX (SP),HL

RET

;

OUTHEX: PUSH AF

RRA

RRA

RRA

RRA

CALL OUTH1

POP AF

OUTH1: PUSH AF

AND 0FH

ADD A,30H

CP 3AH

JR C,OUTH2

ADD A,7

OUTH2: CALL COA

POP AF

RET

OUTHL: PUSH AF

LD A,H

CALL OUTHEX

LD A,L

CALL OUTHEX

POP AF

RET

;--------------------------------------------

; HARDWARADRESSEN PERIPHERIE-SCHALTKREISE

CABIT EQU 1000H ; BITSTRING

CABYTE EQU    0F00H ; BYTES

PIOAST EQU 086H ;PORT A STEUER

PIOBST EQU 087H ;PORT B STEUER

PIOADA EQU 084H ;PORT A DATEN

PIOBDA EQU 085H ;PORT B DATEN

SIOD EQU 088H ; DATEN     / MONITORSCHNITTSTELLE / 

A KANAL SIO

SIOC EQU 08AH ; STEUERWORT

SIOBD EQU 089H ; DATEN     / HOSTSCHNITTSTELLE / B 

KANAL SIO

SIOBC EQU 08BH ; STEUERWORT

CTC EQU 080H ; CTC ADRESSE (BAUDRATE FUER SIO A)

CTC1 EQU 081H ; CTC ADRESSE (BAUDRATE FUER SIO B)

CTC2 EQU 082H

CTC3 EQU 083H

SIOVEK  EQU 00H ; INTERRUPTVEKTOR SIO

CTCVEK EQU 10H

IREG EQU 0FFH ; INTERRUPTREGISTER

CR EQU 0DH

LF EQU 0AH ; LINE FEED

ESC EQU 01BH 

PIOAD EQU 94H ; PIO AUF 3. LP

PIOAC EQU 96H

PDATOU EQU 90H     

BITB EQU 0CFH ; BITBETRIEB

PIOBD EQU     95H

PIOBC EQU 97H

OUTB EQU 0FH ; BYTE AUSGABE

CFDC EQU 9CH

DFDC EQU 9DH

BFDC EQU 9EH

DMAC EQU 8CH

BIOS EQU 0E600H

LST EQU BIOS+0FH

;--------------------

; SPEICHERZELLEN WERTE

ARG0: DEFW 0

ARG1: DEFW 0

ARG2: DEFW 0 ; ENDE IM SPEICHER

ARG3: DEFW 0

PIO: DEFB 0 ; PIO URSPRUNGSBYTE

STRING: DS 20

ENDMC: DEFS 110H ; STACKRAUM

;-------------------------------------------------

end

8


